

NAG Library Routine Document

G05SJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SJF generates a vector of pseudorandom numbers taken from a gamma distribution with parameters a and b .

2 Specification

```
SUBROUTINE G05SJF(N, A, B, STATE, X, IFAIL)
  INTEGER          N, STATE(lstate), IFAIL
  double precision A, B, X(N)
```

3 Description

The gamma distribution has PDF (probability density function)

$$f(x) = \frac{1}{b^a \Gamma(a)} x^{a-1} e^{-x/b} \quad \text{if } x \geq 0; \quad a, b > 0$$

$$f(x) = 0 \quad \text{otherwise.}$$

One of three algorithms is used to generate the variates depending upon the value of a :

- (i) if $a < 1$, a switching algorithm described by Dagpunar (1988) (called G6) is used. The target distributions are $f_1(x) = cax^{a-1}/t^a$ and $f_2(x) = (1-c)e^{-(x-t)}$, where $c = t/(t + ae^{-t})$, and the switching parameter, t , is taken as $1 - a$. This is similar to Ahrens and Dieter's GS algorithm (see Ahrens and Dieter (1974)) in which $t = 1$;
- (ii) if $a = 1$, the gamma distribution reduces to the exponential distribution and the method based on the logarithmic transformation of a uniform random variate is used;
- (iii) if $a > 1$, the algorithm given by Best (1978) is used. This is based on using a Student's t -distribution with two degrees of freedom as the target distribution in an envelope rejection method.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SJF.

4 References

Ahrens J H and Dieter U (1974) Computer methods for sampling from gamma, beta, Poisson and binomial distributions *Computing* **12** 223–46

Best D J (1978) Letter to the Editor *Appl. Statist.* **27** 181

Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the number of pseudorandom numbers to be generated.
Constraint: $N \geq 0$.

- 2: *A* – **double precision** *Input*
 On entry: *a*, the parameter of the gamma distribution.
 Constraint: $A > 0.0$.
- 3: *B* – **double precision** *Input*
 On entry: *b*, the parameter of the gamma distribution.
 Constraint: $B > 0.0$.
- 4: STATE(*lstate*) – INTEGER array *Communication Array*
 Note: *lstate* = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.
 On entry: contains information on the selected base generator and its current state.
 On exit: contains updated information on the state of the generator.
- 5: X(N) – **double precision** array *Output*
 On exit: the *n* pseudorandom numbers from the specified gamma distribution.
- 6: IFAIL – INTEGER *Input/Output*
 On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 0$.

IFAIL = 2

On entry, $A \leq 0.0$.

IFAIL = 3

On entry, $B \leq 0.0$.

IFAIL = 4

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints a set of five pseudorandom numbers from a gamma distribution with parameters $a = 5.0$ and $b = 1.0$, generated by a single call to G05SJF, after initialization by G05KFF.

9.1 Program Text

```
*      G05SJF Example Program Text
*      Mark 22 Release. NAG Copyright 2006.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSTATE, MSEED, N
      PARAMETER        (MSTATE=633,MSEED=1,N=5)
*      .. Local Scalars ..
      DOUBLE PRECISION A, B
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, SUBID
*      .. Local Arrays ..
      DOUBLE PRECISION X(N)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*      .. External Subroutines ..
      EXTERNAL         G05KFF, G05SJF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05SJF Example Program Results'
      WRITE (NOUT,*)
*      Set the distribution parameters A and B
      A = 5.0D0
      B = 1.0D0
*      Initialize the seed to a repeatable sequence
      SEED(1) = 1762543
*      GENID and SUBID identify the base generator.
      GENID = 1
      SUBID = 1
*      Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*      Generate the variates
      IFAIL = 1
      CALL G05SJF(N,A,B,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*      Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
      20 CONTINUE
*
99999  FORMAT (1X,F10.4)
99998  FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997  FORMAT (1X,' ** G05SJF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

G05SJF Example Program Results

5.0702
6.1337
3.1018
3.9863
4.9648
